Singularity Loci of 3-RRR Planar Parallel Manipulators Special Case: Distal and Proximal Links of Equal Length

Ilian A. Bonev and Clement M. Gosselin October 24, 2000

> restart: with(linalg):

Part I: Deriving the Expre ssion for the Jacobian Matrix

> l1 := 1: l2 := 1:
  a1 := 0: b1 := 0: c1 := 0: d1 := 0:
  d2 := b2:

1. Unit Vectors Along Distal Links

> for i from 1 to 3 do
    p||i := (rho||i + l1^2 - l2^2)/(2*l1):
    Gamma||i := rho||i - p||i^2:
    sinT||i := (p||i*(Y+b||i) + (X+a||i)*Delta||i)/rho||i:
    cosT||i := (p||i*(X+a||i) - (Y+b||i)*Delta||i)/rho||i:
    n||i := [X + a||i - l1*cosT||i, Y + b||i - l1*sinT||i]:
  od:

2 . Vectors E*si

> for i from 1 to 3 do
    es||i := [-d||i, c||i];
  od:

3 . Jacobian Matrix

> J := matrix(3,3,[n1[1],n1[2],n1[1]*es1[1] + n1[2]*es1[2],
                   n2[1],n2[2],n2[1]*es2[1] + n2[2]*es2[2],
                   n3[1],n3[2],n3[1]*es3[1] + n3[2]*es3[2]]):

Part II: Obtaining the Singularity Loci

> Eq := simplify(numer(det(J))):

The denominator of det(J) is simply 8*rho1*rho2*rho3.

1 . Getting Rid of The First Radical

After rearranging Eq and rasing to power two, we cancel the common factors rho1, rho2, and rho3. Note that these comprise in fact the numerator of the determinant of J.

> Eq1 := expand( coeff(Eq,Delta1,1)^2*factor(Gamma1/rho1) -
    expand( factor(coeff(Eq,Delta1,0)/rho1)^2 * rho1 ) ):

If we try to substitute Delta2^2 with Gamma2 and Delta3^2 with Gamma3 directly in Eq1, the expression for the latter becomes too big to handle. That is why, we partition Eq1 in 9 different parts and perform the substitutions manually.

> E11 := factor(coeff(coeff(Eq1,Delta2,0),Delta3,0)/(rho2*rho3)):
  E12 := factor(coeff(coeff(Eq1,Delta2,0),Delta3,1)/(rho2*rho3)):
  E13 := factor(coeff(coeff(Eq1,Delta2,1),Delta3,0)/(rho2*rho3)):
  E14 := factor(coeff(coeff(Eq1,Delta2,1),Delta3,1)/(rho2*rho3)):
  E15 := factor(coeff(coeff(Eq1,Delta2,0),Delta3,2)/rho2)*factor(Gamma3/rho3):
  E16 := factor(coeff(coeff(Eq1,Delta2,2),Delta3,0)/rho3)*factor(Gamma2/rho2):
  E17 := factor(coeff(coeff(Eq1,Delta2,1),Delta3,2)/rho2)*factor(Gamma3/rho3):
  E18 := factor(coeff(coeff(Eq1,Delta2,2),Delta3,1)/rho3)*factor(Gamma2/rho2):
  E19 := coeff(coeff(Eq1,Delta2,2),Delta3,2)*
         factor(Gamma2/rho2)*factor(Gamma3/rho3):
> Eq1 := E11 + E12*Delta3 + E13*Delta2 + E14*Delta2*Delta3 + E15 + E16 +
         E17*Delta2 + E18*Delta3 + E19:

2 . Getting Rid of The Second Radical

> E21 := E13 + E17:
  E22 := E14:
  E23 := E11 + E15 + E16 + E19:
  E24 := E12 + E18:
> Eq2 := E21^2*Gamma2 + E22^2*Gamma2*Gamma3 + 2*E21*E22*Gamma2*Delta3 -
         (E23^2 + Eq24^2*Gamma3 + 2*E23*Eq24*Delta3):

3 . Getting Rid of The Third Radical

> Eq3 := (2*E21*E22*Gamma2 - 2*E23*E24)^2*Gamma3 -
         (E21^2*Gamma2 + E22^2*Gamma2*Gamma3 - E23^2 - E24^2*Gamma3)^2:

Finally, we substitute the expressions fot rho1, rho2, and rho3 to obtain the desired polynomial.

> Eq3f := subs({rho1=X^2+Y^2, rho2=(X+a2)^2+(Y+b2)^2, rho3=(X+a3)^2+(Y+b3)^2},
  Eq3):
> degree(Eq3f,X), degree(Eq3f,Y), degree(Eq3f,[X,Y]);

, , 48 48 48 The degree of the polynomial in Eq3f seems to be 48 since the latter is not expanded. However, if we simplify the coefficients of all terms of degree more than 42, we can see that they are all zeros.

> for i from 0 to 48 do
    for j from max(0,43-i) to (48-i) do
      print(simplify(coeff(coeff(Eq3f,X,i),Y,j)));
    od;
  od;

Surprisingly, the coefficients of all terms of degree less than 8 are also zeros.

> for i from 0 to 7 do
    for j from 0 to (7-i) do
      print(simplify(coeff(coeff(Eq3f,X,i),Y,j)));
    od;
  od;

Therefore, the degree of the polynomial, in fact fewnomial, in Eq3 is 42.